SIWAREX FTA称重模块用于耐火材料生产
张胜  时吉敏
某国外知名耐火材料公司在中国建立工厂，由于考虑到对配方的高度保密性以及对配料精度的高要求，因此在整个配料系统中使用了西门子公司的SIWAREX FTA称重模块。我有幸参与了整个项目的设计实施。
该项目配料系统分不定型耐火材料及CCN耐火材料两大部份。其中不定型耐火材料分为酸中性耐火材料和碱性耐火材料，CCN耐火材料又分为AG和ZG两部份。共使用SIWAREX FTA称重模块19个。

不定型耐火材料广泛应用于高炉等，其中酸中性耐火材料有约35种原料配比组成。碱性耐火材料有约14种原料配比组成。下图为不定型配料系统结构：

CCN耐火材料主要用于生产连铸三大件，其中AG约22种原料配比组成。ZG约10种原料配比组成。值得一提的是，100Kg的CCN原料制成的成品价值等同于一辆小轿车。下图为CCN配料系统结构：

整个配料系统要求配料精度较高，因此选择了SIWAREX FTA称重模块+HBM称重传感器的配置。SIWAREX FTA称重模块有采样周期短，AD精度高，集成了较多称重方面经验参数，自己能独立运算等优势。
由于配料系统所配物料种类较多。每个螺旋有多级速度，螺旋下料口还装有控制精确给料的气动阀。部份下料装置为多级振动给料机及微量称等。整个系统称量部份I/O约2000点。在系统自动配料时，为控制料仓压力变化对配料精度产生的影响，配料系统还和上面的料仓自动加料系统有复杂的连锁关系（任何料仓配料时不能加料，加料时不能配料），控制比较复杂。SIWAREX FTA称重模块将称重功能集成到PLC中，复杂控制比专业称重仪表更加柔性一些。
用户配方数量相对所配物料种类更多，达到数百种。而且对配方的高度保密性要求极为严苛，每种配方所需物料种类及配比操作人员不能知道，（很多国外运来物料都使用了代码表示）。在前期技术交流时对方特意关照不能让现场操作人员根据重量等数据侧面了解配方。对配方保密性的要求是甲方决定使用SIWAREX FTA称重模块最直接的原因。
虽然国内诸如洛阳耐火材料研究院等亦拥有先进的技术，但高端产品始终停留在实验室阶段，规模生产无法长期保障稳定的质量。而该配料系统内部写了严格的检测程序。一但某种物料超过了配料合格允许百分比，则启动容错程序对其进行补料。如补料也无法达到要求，则几吨物料将自动进入废品仓。这也客观要求了控制系统的高稳定性及高可靠性。西门子的SIWAREX FTA称重模块在一年多的工作期间经受了考验。
由于整个工厂使用的是三菱PLC系统，西门子SIWAREX FTA称重模块只是挂在几个315-2DP上作为从站使用。西门子与三菱PLC一次只能交换16字数据。但一次配方传输就有数百个字节，而且还有称重模块的检测及控制输出布尔量若干。因此，在编写PLC程序时花了较多心思，建议大家在有大量底层数据交换时尽量采用同一厂家设备，以避免不必要的麻烦。
在使用SIWAREX FTA称重模块编程时，借用了它提供的一个FB41程序，但总体感觉该程序还不够完善，西门子公司应多将中国目前较普及的称重功能写几个专用功能块。SIWAREX FTA称重模块的中文资料怀疑非称重专业人士翻译。少量词汇需参考英文资料才能理解。
该工厂06年底投产至今，系统运行良好。配料系统称量精度优于OIMLⅢ。配料精度大部份达到1‰，少量特殊物料及微量称达到3‰，符合系统设计要求。该工厂产品主要外销，国内仅供自己同一公司使用。
最后，留现场图片及一小段SIWAREX FTA称重模块程序，以供参考探讨之用：

1#FTA
      CALL  "SIWA_FTA" , "CF_A_FB41"    //DB202为FTA背景数据块

       ADDR        :=288                //为FTA物理地址

       DB_SCALE    :=201                //DB201为FTA定标数据块

       DB_VECTOR   :=11                 //DB11为FTA矢量数据块

       CMD_IN      :="CF_A_UDT12".i_CMD_INPUT    //DBW6里设定数据来控制称应该作什么动作

       SIM_VAL     :="CF_A_UDT12".r_SIM_VALUE    //模拟称重量值（相当于用电位器模拟重量）

       ANA_OUT     :="CF_A_UDT12".r_ANALOG_OUT_VALUE    //强制定义FTA模拟输出值4－20MA

       DO_FORCE    :="CF_A_UDT12".b_DIG_OUTPUT_FORCE    //强制定义FTA数字输出

       TRANSITION  :="CF_A_UDT12".b_TRANSITIONS    //自动称重流程中暂停和再启动信号

       CMD_INPR    :="CF_A_UDT12".bo_CMD_IN_PROGRESS    //通知用户，正在处理一个命令

       CMD_FOK     :="CF_A_UDT12".bo_CMD_FINISHED_OK    //通知用户，命令已经成功处理一个命令

       CMD_ERR     :="CF_A_UDT12".bo_CMD_ERR    //通知用户，命令还没有被执行

       CMD_ERR_C   :="CF_A_UDT12".b_CMD_ERR_CODE    //通知用户，命令被执行，但有错误

       REF_COUNT   :="CF_A_UDT12".b_INFO_REFRESH_COUNT    //当前输出值周期性通过外围FB读取

       PROC_VAL1   :="CF_A_UDT12".r_PROCESS_VALUE1    //输出称的静重（在FTA软件里设定过程值1输出当前秤重量值）

       PROC_VAL2   :="CF_A_UDT12".dw_PROCESS_VALUE2    //输出称的控制状态（在FTA软件里设定过程值2输出秤控制输出）

       SC_STATUS   :="CF_A_UDT12".dw_SCALE_STATUS    //非自动称重仪器状态NAWI

       ERR_MSG     :="CF_A_UDT12".bo_ERR_MSG    //通知用户错误信息类型

       ERR_MSG_TYPE:="CF_A_UDT12".b_ERR_MSG_TYPE    //通知用户那重信息代码

       ERR_MSG_C   :="CF_A_UDT12".b_ERR_MSG_CODE    //通知用户信息编号

       FB_ERR      :="CF_A_UDT12".bo_FB_ERR    //功能快本身错误信息

       FB_ERR_C    :="CF_A_UDT12".b_FB_ERR_CODE    //FTA错误信息编号

       START_UP    :="CF_A_UDT12".bo_START_UP_IN_PROGRESS    //输出一个让FTA和CPU同步化的位

       CMD_EN      :="CF_A_UDT12".bo_CMD_ENABLE    //触发FTA执行称重操作

       ERR_MSG_Q   :="CF_A_UDT12".bo_ERR_MSG_QUIT    //故障复位
秤1的AWI状态(从FTA模块读出当前重量值）
      L     "CF_A_UDT12".r_PROCESS_VALUE1

      L     1.000000e+002

      *R                                //从FTA读出的当前重量值乘100，已提高传输精度

      RND                               //将乘100后的当前重量值取整

      T     QW    54                    //传到PROFIBUS传输数据区的QW54里
I13.2:CF_1#秤A物料启动

查看设定值是否大于零，如果大于则启动加料程序（Q74.2）

给定设定量，慢加量，提前量

//将设定值DB200.DBD0给定到FTA_DB201.DBD526里

//将提前值DB200.DBD4给定到FTA_DB201.DBD538里

//将慢加值DB200.DBD8给定到FTA_DB201.DBD542里
      A     I     13.2

      FP    "CF中转位".M501_8

      =     L      6.0

      A     L      6.0

      A(    

      L     "CF_A".CF_A_SET12_1

      L     0.000000e+000

      >R    

      )     

      =     L      6.1

      A     L      6.1

      JNB   _001

      L     "CF_A".CF_A_SET12_1

      T     "CF_A_UDT12".s_DOSING_SET_POINT.r_SET_POINT_VALUE

_001: NOP   0

      A     L      6.1

      JNB   _002

      L     "CF_A".CF_A_SET2_FLIGHT_1

      T     "CF_A_UDT12".s_DOSING_PARA.r_IN_FLIGHT_VALUE

_002: NOP   0

      A     L      6.1

      JNB   _003

      L     "CF_A".CF_A_SET2_FINE_1

      T     "CF_A_UDT12".s_DOSING_PARA.r_FINE_VALUE

_003: NOP   0

      A     L      6.1

      BLD   102

      S     Q     74.2

      A     L      6.0

      BLD   102

      R     Q     74.3

      A     L      6.0

      BLD   102

      R     Q     74.5
FTA执行命令完成信号复位两个称重运行信号

DB201.DBX26.6是FTA称重循环结束

Q74.6:CF_2#秤A物料称量运行复位

Q75.0：CF_2#秤B物料称量运行复位
      A     DB201.DBX   26.6

      FP    "CF中转位".M501_1

      R     Q     74.2

      R     Q     74.4
####1

      A     I     13.2

      L     S5T#1S

      SD    T    155
I13.2:CF_1#秤A物料启动

<=0则1#称未设定重量，称重完成（Q74.3）

Q74.2:CF_1#秤A物料称量运行  

Q74.3:CF_1#秤A物料称量完成

DB201.DBX26.6是FTA称重循环结束，启动CF_1#B物料称。
      A     T    155

      A(    

      L     "CF_A".CF_A_SET12_1

      L     0.000000e+000

      <=R   

      )     

      FP    "CF中转位".BB

      O(    

      A     Q     74.2

      FN    M    699.1

      )     

      S     Q     74.3
秤稳定后延时启动FTA

      A     DB201.DBX   26.0

      L     S5T#2S

      SD    T    151
启动秤重程序

Q74.2:CF_1#秤A物料称量运行  

//651在FTA命令清单6-118页内，将651写入DB201.DBW44里，FTA开始运行。

//DB201.DBX46.0是将触发命令输入FTA
      A     Q     74.2

      A     T    151

      FP    "CF中转位".M501_0

      =     L      6.0

      A     L      6.0

      JNB   _004

      L     651

      T     "CF_A_UDT12".s_CMD2.i_CMD2_Code

_004: NOP   0

      A     L      6.0

      BLD   102

      S     "CF_A_UDT12".s_CMD2.bo_CMD2_Trigger
非称重停止状态下先停止FTA模块（参考第六章命令解释105代码）
      A     I     13.2

      FN    "CF中转位".M501_5

      =     L      6.0

      A     L      6.0

      JNB   _005

      L     105

      T     "CF_A_UDT12".s_CMD1.i_CMD1_Code

_005: NOP   0

      A     L      6.0

      BLD   102

      S     "CF_A_UDT12".s_CMD1.bo_CMD1_Trigger
非称重停止状态下再中止FTA模块（参考第六章命令解释108代码）

DB201.DBX28.5是DR30里的称重停止信号
      A     DB201.DBX   28.5

      FP    "CF中转位".M501_6

      =     L      6.0

      A     L      6.0

      JNB   _006

      L     108

      T     "CF_A_UDT12".s_CMD1.i_CMD1_Code

_006: NOP   0

      A     L      6.0

      BLD   102

      S     "CF_A_UDT12".s_CMD1.bo_CMD1_Trigger
非称重停止状态下停1＃2＃螺旋

      A     I     13.2

      FN    "CF中转位".M501_7

      R     Q     74.2

      R     Q     74.4

      R     Q     74.3

      R     Q     74.5
FC30用来给启动FTA模块区分启动命令等级

最高等级是sCmd_1

中等级是sCmd_2

低等级是sCmd_3
      CALL  "Fc Execute command"

       iCmdInput      :="CF_A_UDT12".i_CMD_INPUT    //三个等级命令值都将输入到FTA的DB201.DBW6里

       boCmdEnable    :="CF_A_UDT12".bo_CMD_ENABLE    //触发命令见书8.5章定标数据库中的分配

       boCmdInProgress:="CF_A_UDT12".bo_CMD_IN_PROGRESS    //触发命令正在执行见书8.5章定标数据库中的分配

       boCmdFinishedOk:="CF_A_UDT12".bo_CMD_FINISHED_OK    //触发命令正常完成见书8.5章定标数据库中的分配

       boCmdErr       :="CF_A_UDT12".bo_CMD_ERR    //触发命令错误见书8.5章定标数据库中的分配

       sCmd_1         :=DB201.DBD40     //最高等级是sCmd_1

       sCmd_2         :=DB201.DBD44     //中等级是sCmd_2

       sCmd_3         :=DB201.DBD48     //低等级是sCmd_3

下第二种物料配方

Q74.3为FTA,CF_1#A称重循环结束，启动CF_1#B物料称。
      A     Q     74.3

      FP    "CF中转位".M501_2

      A(    

      L     "CF_A".CF_A_SET1_1

      L     0.000000e+000

      >R    

      )     

      =     L      6.0

      A     L      6.0

      JNB   _007

      L     "CF_A".CF_A_SET1_1

      T     "CF_A_UDT12".s_DOSING_SET_POINT.r_SET_POINT_VALUE

_007: NOP   0

      A     L      6.0

      JNB   _008

      L     "CF_A".CF_A_SET1_FLIGHT_1

      T     "CF_A_UDT12".s_DOSING_PARA.r_IN_FLIGHT_VALUE

_008: NOP   0

      A     L      6.0

      JNB   _009

      L     "CF_A".CF_A_SET1_FINE_1

      T     "CF_A_UDT12".s_DOSING_PARA.r_FINE_VALUE

_009: NOP   0

      A     L      6.0

      BLD   102

      S     Q     74.4
Q74.3:CF_1#秤A物料完成 CF_1#秤B物料启动

<=0则CF_1#B称未设定重量，称量完成（Q74.5）

Q74.4:CF_1#秤B物料称量运行  

Q74.5:CF_1#秤B物料称量完成

DB201.DBX26.6是FTA称重循环结束，CF_1#称称重配料完成。
      A     Q     74.3

      A(    

      L     "CF_A".CF_A_SET1_1

      L     0.000000e+000

      <=R   

      )     

      O(    

      A     Q     74.4

      FN    "CF中转位".M501_9

      )     

      S     Q     74.5
第二次启动秤重程序

//651在FTA命令清单6-118页内，

//下一行是将启动命令输入FTA
      A     Q     74.4

      A     T    151

      FP    "CF中转位".M501_4

      =     L      6.0

      A     L      6.0

      JNB   _00a

      L     651

      T     "CF_A_UDT12".s_CMD2.i_CMD2_Code

_00a: NOP   0

      A     L      6.0

      BLD   102

      S     "CF_A_UDT12".s_CMD2.bo_CMD2_Trigger
以下是秤状态读取并传送到三菱，对应三菱W806
秤快加状态读取

//DB201.DBX28.1为FTA DR30里粗加信号9 

具体双字对应字节排列方式详见《S7-300400语句表编程中

参考手册2-7页。
      A     DB201.DBX   28.1

      =     Q     76.0
秤慢加状态读取

//DB201.DBX28.2为FTA 

DR30里粗加信号10具体双字对应字节排列方式详见《S7-300400语句表编程中

参考手册2-7页。
      A     DB201.DBX   28.2

      =     Q     76.1
秤后续给料激活

//DB201.DBX28.0为FTA DR30里粗加信号8
      A     DB201.DBX   28.0

      =     Q     76.2
称重结束信号

//DB201.DBX26.7为FTA DR30里粗加信号30
      A     DB201.DBX   26.6

      =     Q     76.3
合格信号输出

//DB201.DBX27.4为FTA DR30里粗加信号20
      A     DB201.DBX   27.4

      =     Q     76.4
不合格信号输出

//DB201.DBX27.7为FTA DR30里粗加信号23
      A     DB201.DBX   27.7

      =     Q     76.5
